
... TIC MARK LABEL FORMAT

PURPOSE
Specifies the format of the labels associated with the major tic marks on subsequent plots.

DESCRIPTION
This command has 2 uses.

1. For linear scales the following formats are available:

ALPHA - the tic labels will be user-supplied (via the TIC MARK LABEL CONTENT command) alphanumeric
strings.

REAL - linear scales are drawn with numeric values.

2. Log scales can be drawn in one of 3 formats. The available formats are:

POWER - the corresponding power of 10 (e.g., 1 for 101) is drawn for the tic label.
REAL - the corresponding real number (e.g., 10 for 101) is drawn for the tic label.
EXPO - the tic label is draw in exponential format (e.g., 101). The tic label must be drawn with a software font in

order for this option to work correctly.

SYNTAX
<prefix>TIC MARK LABEL FORMAT <format>
where <prefix> is one of the following:

X1 refers to the bottom horizontal side;
X2 refers to the top horizontal side;
X refers to both horizontal sides;
Y1 refers to the left vertical side;
Y2 refers to the right vertical side;
Y refers to both vertical sides;
no prefix refers to all 4 sides;

and  <format> is one of ALPHA/REAL/POWER/EXPO as described in the DESCRIPTION section above.

EXAMPLES
TIC MARK LABEL FORMAT POWER
TIC MARK LABEL FORMAT
XTIC MARK LABEL FORMAT ALPHA

NOTE
A ...TIC MARK LABEL FORMAT command with no arguments reverts the setting to default; thus X1TIC MARK LABEL FORMAT
with no arguments reverts the bottom horizontal tic label to the default format. A ...TIC MARK LABEL FORMAT command with no
prefix refers to all 4 sides; thus TIC MARK LABEL FORMAT REAL sets tic mark labels to real for all 4 frame lines. TIC MARK
LABEL FORMAT with no prefix and no arguments reverts the tic label format settings on all 4 sides to default.

DEFAULT
The tic mark labels for linear scales are real numbers. The tic mark labels for log scales are drawn with the POWER format.

SYNONYMS
TIC LABEL FORMAT, TICS LABEL FORMAT, and TICS MARK LABEL FORMAT are synonyms for TIC MARK LABEL
FORMAT, as in TIC LABEL FORMAT REAL and X2TIC LABEL FORMAT REAL.

RELATED COMMANDS
TIC MARK LABEL = Specifies whether or not tic mark labels or drawn.
TIC MARK LABEL ANGLE = Sets the angle for tic labels.
TIC MARK LABEL CASE = Sets case for tic labels.
TIC MARK LABEL COLOR = Sets the color for tic labels.
TIC MARK LABEL CONTENTS = Specify alphanumeric values for tic labels.
TIC MARK LABEL DECIMALS = Sets the number of decimals for tic labels.
TIC MARK LABEL DIRECTION = Sets the direction (horizontal/ vertical) for TIC MARK labels.
TIC MARK LABEL DISPLACEMENT= Sets the distance between tic labels and the frame.
TIC MARK LABEL FONT = Sets the font for tic labels.

TIC MARK LABEL HW = Sets the height and width for tic labels.
TIC MARK LABEL JUSTIFICATION = Sets the justification for tic labels.
TIC MARK LABEL SIZE = Sets the height for tic labels.
TIC MARK LABEL THICKNESS = Sets the line width for tic labels.

APPLICATIONS
Log scales, presentation graphics

IMPLEMENTATION DATE
88/2

PROGRAM 1
LET Y = DATA 392 341 307 203 115 59 38 32 29 28
LET N = NUMBER Y
LET X = SEQUENCE 1 1 N
LINES BLANK ALL
BAR ON ALL
BAR WIDTH .5 ALL
BAR DIMENSION 3 ALL; BAR FILL ONTS ALL
XLIMITS 1 10
XTIC OFFSET 1 1
YLIMITS 0 500
MAJOR XTIC MARK NUMBER 10
MINOR XTIC MARK NUMBER 0
X1TIC MARK LABEL FORMAT ALPHA
X1TIC MARK LABEL CONTENT Apollo Sun HP DEC SYM SGI IBM DG TI Xerox
X1TIC MARK LABEL CASE ASIS
LEGEND 1 PRESENTATION GRAPHICS
LEGEND 2 3-D BAR HISTOGRAM
TITLE DEMONSTRATE TIC LABEL FORMAT; TITLE SIZE 5
PLOT Y X X

Apollo Sun HP DEC SYM SGI IBM DG TI Xerox

0

100

200

300

400

500

DEMONSTRATE TIC LABEL FORMAT

PRESENTATION GRAPHICS

3-D BAR HISTOGRAM



... TIC MARK LABEL FORMAT Plot Control Commands

4-286 March 17, 1997 DATAPLOT Reference Man ual

PROGRAM 2
SERIAL READ Y

 60. 2042. 2111. 1684. 3888. 1858. 11379. 17560. 39287.
 4382. 113159. 175108. 273291. 400186. 581243. 811568.
1121004. 1506550. 2002767. 2611612. 3369180.

END OF DATA
SERIAL READ X

 0. 1. 32. 243. 1024. 3125. 7776. 16807. 32768. 59049.
100000. 161051. 248832. 371293. 537824. 759375. 1048576.
1419857. 1889568. 2476099. 3200000.

END OF DATA
.
TITLE SIZE 3
TIC LABEL SIZE 2.5
MULTIPLOT 2 2; MULTIPLOT CORNER COORDINATES 0 0 100 100
TITLE LINEAR SCALE
PLOT Y X
TITLE LOG SCALES (POWER FORMAT)
LOG ON
LET X = 0.0001 SUBSET X <= 0
PLOT Y X
TITLE LOG SCALE (REAL FORMAT)
TIC LABEL FORMAT REAL
PLOT Y X
TITLE LOG SCALE (EXPONENTIAL FORMAT)
TIC LABEL FORMAT EXPONENTIAL
TIC LABEL FONT SIMPLEX
PLOT Y X
END OF MULTIPLOT

